Maximaを用いた数式処理2

1

数式処理	厳密解 （文字処理）	・数式の展開，因数分解 ・方程式の求解 ・行列演算 ・微分，積分 ・常微分方程式の解析解

方程式の記憶

方程式の記憶	方程式名:方程式の内容;
$x^2-1=0$	eq:x^2-1=0;

極限値（Limit）

2

関数f(x)がaにおいて連続の場合

$\lim_{x\to a} f(x)$	limit(f(x),x,a)

$\lim_{x\to 0}\frac{\sin 3x}{x}=3$　　limit(sin(3*x)/x,x,0)

$\lim_{x\to +\infty}\frac{2x+1}{3x+2}=\frac{2}{3}$　　limit((2*x+1)/(3*x+2),x,inf)

関数f(x)がaにおいて不連続の場合

limit(f(x),x,a,dir)	dir:接近方向

$\lim_{x\to a+0}\frac{1}{x-a}=+\infty$　　limit(1/(x-a),x,a,plus)

$\lim_{x\to a-0}\frac{1}{x-a}=-\infty$　　limit(1/(x-a),x,a,minus)

微分 (Differentiation)

3

1次微分　$\frac{d}{dx}f(x)$	diff(f(x),x)

$\frac{d}{dx}\cos x=-\sin x$　　diff(cos(x),x);

$\frac{d}{dx}x^n=nx^{n-1}$　　diff(x^n,x);

高次微分　$\frac{d^n}{dx^n}f(x)$	diff(f(x),x,n)

$\frac{d^2}{dx^2}x^2\sin x=2\sin x+4x\cos x-x^2\sin x$

diff(x^2*sin(x),x,2)

積分 (Integration)

4

不定積分　$\int f(x)dx$	integrate(f(x),x)

$\int\frac{1}{1+x}dx=\log(x+1)$　　integrate(1/(1+x^3),x)

Maximaでは積分定数は0に設定される

定積分　$\int_a^b f(x)dx$	integrate(f(x),x,a,b)

$\int_0^1\frac{1}{1+x^2}dx=\frac{\pi}{4}$　　integrate(1/(1+x^2),x,0,1)

テーラー展開 (Taylor Expansion)

5

$f(a+x)\approx f(a)+xf'(a)+\frac{x^2}{2}f''(a)+\frac{x^3}{3!}f^{(3)}(a)+\cdots+\frac{x^n}{n!}f^{(n)}(a)$
taylor(f(x),x,a,n)

$\sin x=x-\frac{x^3}{3!}+\frac{x^5}{5!}-\frac{x^7}{7!}+\frac{x^9}{9!}+\cdots$

0次から10次までのテーラー展開

taylor(sin(x),x,0,10)

未知関数の展開

入力例	taylor(f(x),x,0,3);
表示結果	f(0) +(at('diff(f(x),x,1),x=0))*x +(at('diff(f(x),x,2),x=0))*x^2/2 +(at('diff(f(x),x,3),x=0))*x^3/6

偏微分 (Partial Differentiation)

6

$\frac{\partial^m\partial^n}{\partial y^m\partial x^n}f(x,y,\cdots)$	diff(f(x,y,..),x,n,y,m)

$\frac{\partial^3}{\partial x\partial y^2}(x^2y^3+y)\longrightarrow 12xy$

diff(x^2*y^3+y,x,1,y,2);

全微分 (Differentiation)

$df(x,y,\cdots)$	diff(f(x,y,..))

$df(x,y)=\frac{\partial f(x,y)}{\partial x}dx+\frac{\partial f(x,y)}{\partial y}dy$　　diff(f(x,y));

$f(x,y)=x^2y^3+y\longrightarrow df=2xy^3dx+(3x^2y^2+1)dy$

diff(x^2*y^3+y);

微分方程式 (Differential Equation)

7

初期条件	atvalue(form,x=x0,y0)
常微分方程式	desolve(deq(y(x),x),y(x))

$\frac{dy(x)}{dx}=a\cdot y(x)\longrightarrow y(x)=y(0)\cdot e^{ax}$

desolve('diff(y(x),x)=a*y(x),y(x));

注

1	微分方程式には，「diff()」ではなく，「'diff()」を用いる．
2	desolve()を用いる場合，y(x)のように依存関係を明示する．すなわち，「'diff(y,x)=a*y」ではなく，「'diff(y(x),x)=a*y(x)」と記述する．
3	atvalue()はdesolve()の前に記述する．
4	desolve()では解けない場合もある．2階以下の微分方程式であれば，それに特化したode2()を用いる．

desolve()の初期条件の設定

8

質量m，バネ定数k，変位x(t)とする．

（1）右のシステムの運動方程式を求めなさい．

（2）運動方程式を解いて，変位x(t)を求めなさい．



$\begin{cases} m\ddot{x}(t)=-kx(t) \\ x(0)=A \\ \dot{x}(0)=0 \end{cases}\longrightarrow x(t)=A\cos(t\sqrt{k/m})$

```
assume(m>0,k>0);
deq: m*'diff(x(t),t,2)=-k*x(t);
atvalue(x(t),t=0,A);
atvalue('diff(x(t),t),t=0,0);
desolve(deq,x(t));
```

連立微分方程式

9

連立微分方程式	desolve([deq1,..],[x1,..])

$\begin{cases} \dot{x}(t)=2x(t)-y(t) \\ \dot{y}(t)=x(t)+2y(t) \\ x(0)=y(0)=1 \end{cases}\longrightarrow \begin{cases} x(t)=e^{2t}(\cos t-\sin t) \\ y(t)=e^{2t}(\sin t+\cos t) \end{cases}$

```
deq1: 'diff(x(t),t)=2*x(t)-y(t);
deq2: 'diff(y(t),t)=x(t)+2*y(t);
atvalue(x(t),t=0,1);
atvalue(y(t),t=0,1);
desolve([deq1,deq2],[x(t),y(t)]);
```

参考

$\begin{pmatrix}\dot{x}(t)\\ \dot{y}(t)\end{pmatrix}=\begin{pmatrix}2 & -1\\ 1 & 2\end{pmatrix}\begin{pmatrix}x(t)\\ y(t)\end{pmatrix}\xrightarrow{A=\begin{pmatrix}2 & -1\\ 1 & 2\end{pmatrix}}\begin{pmatrix}x(t)\\ y(t)\end{pmatrix}=e^{At}\begin{pmatrix}x(0)\\ y(0)\end{pmatrix}$

連立微分方程式

$$\begin{cases} \dfrac{df(x)}{dx} = \dfrac{dg(x)}{dx} + \sin x \\ \dfrac{d^2 g(x)}{dx^2} = \dfrac{df(x)}{dx} - \cos x \\ f(0) = a \\ g(0) = b, g'(0) = c \end{cases} \longrightarrow \begin{cases} f(x) = c(e^x - 1) + a \\ g(x) = \cos x + c(e^x - 1) + b - 1 \end{cases}$$

```
deq1: 'diff(f(x),x)='diff(g(x),x)+sin(x);
deq2: 'diff(g(x),x,2)='diff(f(x),x)-cos(x);
atvalue(f(x),x=0,a);
atvalue(g(x),x=0,b);
atvalue('diff(g(x),x),x=0,c);
desolve([deq1,deq2],[f(x),g(x)]);
```

2階以下の常微分方程式

方程式の解	ode2(deq(y",y',y,x),y,x)
1階の初期条件	ic1(eq,x0,y(x0))
2階の初期条件	ic2(eq,x0,y(x0),y'(x0))
境界条件	bc2(f(x),x1,y(x1),x2,y(x2))

$$x^2 \frac{dy}{dx} + 3xy = \frac{\sin x}{x} \longrightarrow y = \frac{C - \cos x}{x^3}$$

```
ode2(x^2*'diff(y,x)+3*y*x=sin(x)/x,y,x)
```

注		
	1	ode2()の場合，微分方程式を「'diff(y,x)=a*y」と記述しても良い．
	2	ode2()の後にic1(),ic2(),bc2()を記述する．

用語 ode: Ordinary Differential Equation （常微分方程式）
ic: Initial Condition （初期条件）
bc: Boundary Condition （境界条件）

ic1()による1階の初期条件

$$\begin{cases} \dfrac{df(x)}{dx} = a \cdot f(x) \\ f(0) = b \end{cases} \longrightarrow f(x) = be^{ax}$$



ic1の設定

```
deq: 'diff(f,x)=a*f;
eq: ode2(deq,f,x);
ic1(eq,x=0,f=b);
```

$$\frac{df(x)}{dx} = a \cdot f(x) \xrightarrow{\text{ode2}} f(x) = Ce^{ax}$$ C:積分定数

$$\xrightarrow{\text{ic1}} f(x) = be^{ax}$$

1階の常微分方程式の解は積分定数が1個

ic2()による2階の初期条件

$$\begin{cases} \ddot{x}(t) + \dot{x}(t) + x(t) = 0 \\ x(0) = 1, \dot{x}(0) = 0 \end{cases} \longrightarrow x(t) = e^{-t/2}\{\frac{\sqrt{3}}{3}\sin(\frac{\sqrt{3}}{2}t) + \cos(\frac{\sqrt{3}}{2}t)\}$$

```
deq: 'diff(x,t,2)+'diff(x,t)+x=0;
eq: ode2(deq,x,t);
ic2(eq,t=0,x=1,'diff(x,t)=0);
```



ic2の設定

$$\ddot{x}(t) + \dot{x}(t) + x(t) = 0$$

$$\xrightarrow{\text{ode2}} x(t) = e^{-t/2}\{k_1 \sin(\frac{\sqrt{3}}{2}t) + k_2 \cos(\frac{\sqrt{3}}{2}t)\}$$ k1,k2 積分定数

$$\xrightarrow{\text{ic2}} x(t) = e^{-t/2}\{\frac{\sqrt{3}}{3}\sin(\frac{\sqrt{3}}{2}t) + \cos(\frac{\sqrt{3}}{2}t)\}$$

2階の常微分方程式の解は積分定数が2個

bc2()による境界条件

$$\begin{cases} y'' + y(y')^3 = 0 \\ y(0) = 1, y(1) = 3 \end{cases} \longrightarrow \frac{y^3 - 10y}{6} = x - \frac{3}{2}$$



bc2の設定

```
deq: 'diff(y,x,2)+y*'diff(y,x)^3=0;
eq: ode2(deq,y,x);
bc2(eq,x=0,y=1,x=1,y=3);
```

$$y'' + y(y')^3 = 0 \xrightarrow{\text{ode2}} \frac{y^3 + 6k_1 y}{6} = x + k_2$$ k1,k2 積分定数

$$\xrightarrow{\text{ic2}} \frac{y^3 - 10y}{6} = x - \frac{3}{2}$$

機械システム

(1) 右のシステムの運動方程式を求めなさい．
(2) 質量m=1，減衰係数b=1，バネ定数k=1，初期位置x(0)=1，初期速度v(0)=0として問題(1)の微分方程式を解きなさい．
(3) 位置と速度をグラフ表示しなさい．



b:粘性係数

(1) $m\ddot{x}(t) + b\dot{x}(t) + kx(t) = 0$

↓ $m = 1, b = 1, k = 1$

(2) $\ddot{x}(t) + \dot{x}(t) + x(t) = 0$

↓ ode2()を利用してx(t)を解く

p.13へ

↓

(3) $v(t) = \dot{x}(t)$ $\xrightarrow{\text{グラフを描画}}$ p.16へ

状態変数（位置と速度）の時間変化


```
x(t):=%e^-(t/2)*
(sin(sqrt(3)*t/2)/sqrt(3)+cos(sqrt(3)*t/2));
v(t):=(-2)*%e^-(t/2)*sin(sqrt(3)*t/2)/sqrt(3);
plot2d([x(t),v(t)],[t,0,10]);
```





状態空間表示（位置と速度の関係）


```
plot2d([parametric,x(t),v(t)],[t,0,20],
        [nticks,200]);
```





入力コマンドの保存

wxMaximaで対話処理形式によるコマンド入力



「File」→「Save as」で保存．
ファイルの種類：「wxMaxima session (*.wxm)」

```
/* [wxMaxima batch file version 1] [ DO NOT EDIT BY HAND! ]*/
/* [ Created by wxMaxima version 0.7.3a ] */

/* [wxMaxima: input   start ] */
a:3*2+5*7;
/* [wxMaxima: input   end   ] */

/* [wxMaxima: input   start ] */
b:3/(2*a+5);
/* [wxMaxima: input   end   ] */

/* [wxMaxima: input   start ] */
c:a+b;
/* [wxMaxima: input   end   ] */

/* Maxima can't load/batch files which end with a comment! */
"Created with wxMaxima"$
```

必要最小限

```
a:3*2+5*7;
b:3/(2*a+5);
c:a+b;
```

/**/はコメント．実行に影響しない．
テキストエディタで修正